

Project 4 - Noise Reduction System

Jun D. Ouyang
Guangliang Wu

Abstract—This project involves the design of a noise reduction system comprised of a pre-emphasis filter and a de-emphasis filter. For this project, we will be designing higher-order active filter circuits using op-amps, and perform a cost analysis of our circuit.

I. OBJECTIVE

A. Pre-Emphasis Filter

- Design a Pre-emphasis filter which the output at frequencies 100 Hz and 10 kHz must be within ± 3 dB of the input $x(t)$, and the output at frequency range 500 Hz to 2 kHz must be bigger than 20 dB.
- Design the circuit with entirely with active circuits, but no inductors.
- Construct the pre-emphasis filter, and then take the gain measurements over the 100 Hz to 10 kHz.
- Covert the gain to dB, then sketch a plot of gain in dB against frequency on a log scale, then compare the plot with the SPICE results.
- Analysis the cost of our circuit.
- Determine the legitamacy of our design.

B. De-Emphasis Filter

- Design a De-emphasis filter which the output at frequencies 100 Hz and 10 kHz must be within ± 3 dB of the input $x(t)$, and the output at frequency range 500 Hz to 2 kHz must be less than -20 dB.
- Design the circuit with entirely with active circuits, but no inductors.
- Construct the de-emphasis filter, and then take the gain measurements over the 100 Hz to 10 kHz.
- Covert the gain to dB, then sketch a plot of gain in dB against frequency on a log scale, then compare the plot with the SPICE results.
- Analysis the cost of our circuit.
- Determine the legitamacy of our design.

C. Noise Reduction System

- Combine Pre-emphasis and De-emphasis filter together.
- Construct the combination of Pre-emphasis and De-emphasis we designed, and then take the gain measurements over the 100 Hz to 10 kHz.
- Cover the gain to dB, then sketch a plot of gain in dB against frequency on a log scale, then compare the plot with the SPICE results.
- Determine the legitamcy of our final design.

II. SPECIFICATION

In designing our circuit we realized two utmost important point in which our circuit has to meet.

- 1) We must obtain approximately 0 dB at 100Hz and 10kHz for both the pre-emphasis and the de-emphasis circuit.
- 2) We must make sure the pre-emphasis filter must be less having at least 20dB more over other values outside the range of 500Hz and 2kHz.
- 3) We must make sure the final cascaded product is within ± 3 dB for our simulation; and due to technical difficulties, ± 5 dB for the experimental output over the range between 100Hz and 10kHz.

To satisfy 1 and 2, we can use high pass and low pass Butterworth filter circuits by calculating the desired transition slope and scale to the final desired frequencies. Then in the case of the pre-emphasis filter, we can cascade the high pass and low pass circuits together to form a band pass filter with desired gain; and as in the case of the de-emphasis, we will also perform the exact same procedure but instead of cascading, we will be summing the low pass and high pass filters to form a band reject filter with exactly the opposite gain of the pre-emphasis stage.

A. Pre-emphasis Stage

In the design of the pre-emphasis stage, we will be cascading a high pass and a low pass filter together and lastly feed through a inverting amplifier to achieve desired gain.

1) *High Pass Filter*: To create the abovementioned circuit, we must first calculate the desired slope. We have done so with the following formula directly from the textbook:

$$n = \frac{\log \left(\frac{\sqrt{10^{-0.1A_s} - 1}}{\sqrt{10^{-0.1A_p} - 1}} \right)}{\log(\omega_p/\omega_s)} \quad (1)$$

Because we want at 100 Hz about 0 dB and at 500 Hz about 20 dB, we can set the cutoff frequency of the high pass filter to be at 500 Hz, where $A_s = -23$ dB, $A_p = -3$ dB, $\omega_s = 100$ Hz, and $\omega_p = 500$ Hz. We can then scale the amplitude by 23 dB (approximately $10^{23/20} = 14.1$ dB to achieve our specification. Using such values and via (1), we obtain the following,

$$n = \frac{\log \left(\frac{\sqrt{10^{-0.1 \times (-23)} - 1}}{\sqrt{10^{-0.1 \times (-3)} - 1}} \right)}{\log(500/100)} \approx 1.64$$

Because we can only have integer value of n , we must either ceil or floor the calculated value. Since the value is closer


to 2 dB, and since the BW filter guaranteed to pass through 500 Hz at -3 dB after frequency scaling, we chose to ceil the calculated n in the hopes to preserve the ± 3 dB boundary of the original input signal at 100 Hz. Therefore, we obtain after ceiling,

$$n = 2 \quad (2)$$

Now we can build the high pass stage with the $n = 2$ (second order) Butterworth high pass filter. The second order BW high pass filter has the s-domain equation:

$$H(s) = \frac{s^2}{s^2 + \frac{2}{R_2 C} s + \frac{1}{R_1 R_2 C^2}} \quad (3)$$

The circuit for the second order BW HP filter is as follows:



Note, in this case, $C_1 = C_2 = C$, and this circuit has cutoff frequency at $\omega_c = \frac{1}{R^2 C_1 C_2} = 500Hz \times 2\pi = 1000 \pi rad/s$. But as off now, its unscaled version has only $\omega_c = 1 rad/s$ as R's and C's are in values of prototype circuit of 1Ω and $1 F$ respectively.

2) *Low Pass Filter*: For the low pass stage of the pre-emphasis filter, we may apply the same concept to the high pass filter to find the suitable BW order for the slope between $\omega_p = 2$ kHz, $\omega_s = 10$ kHz, $A_p = -3$ dB, and $A_s = -23$ dB. We apply the low-pass BW order formula below,

$$n = \frac{\log \left(\frac{\sqrt{10^{-0.1A_s}} - 1}{\sqrt{10^{-0.1A_p}} - 1} \right)}{\log(\omega_s/\omega_p)} \quad (4)$$

Obtaining,


$$n = \frac{\log \left(\frac{\sqrt{10^{-0.1 \times (-23)}} - 1}{\sqrt{10^{-0.1 \times (-3)}} - 1} \right)}{\log(10000/2000)} \approx 1.64$$

Therefore, we will also choose second order BW filter as we did with the high pass stage. The second order low-pass BW filter has the following s-domain frequency response:


$$H(s) = \frac{1/(R^2 C_1 C_2)}{s^2 + \frac{2}{R C_1} s + \frac{1}{R^2 C_1 C_2}} \quad (5)$$

The circuit for the second order BW LP filter is as follows:

$$\text{With } \omega_c = \frac{1}{R^2 C_1 C_2}.$$



3) *Gain stage*: The pre-emphasis must employ a gaining stage to have its values above 0 dB. Using an inverting op-amp configuration in such case proves to be most direct approach. It has simple transfer function of $H(j\omega) = -\frac{R_f}{R_i}$. Its schematics is shown below:



By cascading the high pass stage, low pass stage and the gain stage in series, will therefore satisfy our specification. Its final product is not shown here as it is easy to imagine. But, as usual, its transfer function, on the other hand, can be calculated by multiplying abovementioned filters' frequency responses. The final transfer function for the pre-emphasis filter is therefore:

$$H(s) = H_{hp}(s) \times H_{lp}(s) \times H_{gain}(s) \quad (6)$$

$$= \frac{s^2}{s^2 + \frac{2}{R_2 C} s + \frac{1}{R_1 R_2 C^2}} \quad (7)$$

$$\times \frac{1/(R^2 C_1 C_2)}{s^2 + \frac{2}{R C_1} s + \frac{1}{R^2 C_1 C_2}} \times \frac{R_f}{R_i} \quad (8)$$

As usual, the center frequency of any bandpass filter is,

$$\omega_o = \sqrt{\omega_{c_{hp}} \omega_{c_{lp}}} = \sqrt{\frac{1}{R_1 R_2 C^2} \times \frac{1}{R^2 C_1 C_2}} \quad (9)$$

To view the full connection of the pre-emphasis filters, see appendix A.

B. De-emphasis Stage


The design of the de-emphasis stage is quite tricky because we cannot simply just rely on the 100Hz and 500Hz or the 2kHz and 10kHz pair, because if we employed frequency scaling to such places, the slope of the frequency response

of the pre-emphasis does not match with that of the de-emphasis. Therefore, we must scale the de-emphasis stage quite differently. But, however, we note that however we might change the cutoff frequency by scaling, the general slope however, does not need to be or mustn't be altered to cancel out the slope of the pre-emphasis filter; and it is precisely this, we can apply the same filter components in the pre-emphasis stage. But this time, instead of building a band-pass filter, we must build a band-reject filter because we need to undo the amplification created by the pre-emphasis filter. Therefore, when we combine the two low pass and high pass filters, we do not cascade them in series but in parallel and have their corresponding output signals summed up to create a band pass filter. In case of exceptions, we can also throw in another inverting op-amp to adjust to the adequate signal level. Therefore, the de-emphasis requires four building blocks: a BW LP filter, a BW HP filter, a summing amplifier, and possibly an inverting op-amp.

1) *Low-pass Stage*: This stage, as its name indicates, undergoes a low-pass filter. It has exactly the same identical structure of the (5) and therefore we will omit its discussion here. The only difference will be its scaling, which will be discussed at length in the simulation section because the calculations done by idealization of the circuit elements is prone to have errors. This low pass stage will actually have K_f scaling the prototype cutoff frequency to around 100Hz.

2) *High-pass Stage*: The high pass stage equivalent to the pre-emphasis high pass filter with equation (3), except having its frequency scaled to having $\pm 3\text{dB}$ at around 10kHz.

3) *Sum Stage*: The summing op-amp is simply the standard unity gain summing op-amp configuration with $H(s) = 1$. Its setup is shown below:



Where $R_1 = R_2 = R_f$ since it is unity gain.

C. Gain stage

We cannot expect any kind of possible disturbance caused by the non-ideality of the op-amp model during simulation and/or experiment. We must therefore allow an extra inverting amplifier in our arsenal to adjust the amplitude of the resulting wave as needed. The inverting op-amp is exactly the same as the one from the pre-emphasis stage. We therefore will not discuss of it.

We can now combine all these elements together into the de-emphasis configuration. It ultimately ends up as the following circuit diagram:

Its transfer function is:

$$H(s) = H_{hp}(s) \times H_{lp}(s) \times H_{gain}(s) \quad (10)$$

$$= \left(\frac{s^2}{s^2 + \frac{2}{R_2 C} s + \frac{1}{R_1 R_2 C^2}} \right) \quad (11)$$

$$+ \frac{1/(R^2 C_1 C_2)}{s^2 + \frac{2}{R C_1} s + \frac{1}{R^2 C_1 C_2}} \times \frac{R_f}{R_i} \quad (12)$$

As usual, the center frequency of any bandreject filter is the same as that of the bandpass filter,

$$\omega_o = \sqrt{\omega_{chp} \omega_{clp}} = \sqrt{\frac{1}{R_1 R_2 C^2} \times \frac{1}{R^2 C_1 C_2}} \quad (13)$$

The only difference between the bandpass of the pre-emphasis stage and the bandreject of the de-emphasis stage is their frequency scaling value and the way components of LP and HP filters connecting together as shown by the above equations. To view the full connection of the de-emphasis filters, see appendix B.

III. SIMULATION

In simulating this nrs circuit, we broke down the entire spice code bit by bit and encompassed automatic expression evaluation to facilitate us from the burden of scaling capacitors or resistors values. Therefore, instead of focusing on how we scale here, we would be able to focus on how we could achieve our specification.

A. The Ingredients

Before we begin explaining our code listings, we will inform this report's reader that our project's source code is available online at GitHub in this following address:

<http://www.github.com/jouyang3/nrs>

Now onto the simulation subject. Because the reusability of the second order low pass and high pass sub-circuit, we were able to parametrize both low pass and high pass sub-circuit to avoid code duplication. The resulting code listing for the low-pass filter is hence as follows in the file lp.cir:

Listing 1. 2nd order BW LP netlist.

```
.SUBCKT lp in out kf=1 r=1
.PARAM km=r
R1 in 1 R=r
R2 1 p R=r
C1 1 out C= ' 1.414/km/ kf '
C2 p 0 C= ' 0.707/km/ kf '
X1 p out out popamp
.ENDS lp
```

Here, we enabled the low pass filter accept scaling argument thus we were able to alter the frequency scaling as needed both by the pre-emphasis and the de-emphasis stage. The expression of C_1 and C_2 enabled automatic scaling when the scaling factor K_f sets in (by default, there's no scaling, hence $K_f = 1$).




Fig. 1. Second order BW LP Circuit.

The circuit with nodes labeled for this LP netlist is reprinted here at the reader's convenience.

We applied the same treatment to the high pass netlist. Here's its netlist:

```
Listing 2. 2nd order BW HP netlist.
.SUBCKT hp in out kf=1 c=1
.PARAM km='1/kf/c'
C1      in      1      C=c
R1      1      out    R='km*0.707'
C2      1      p      C=c
R2      p      0      R='km*1.414'
X1      p      out    out    popamp
.ENDS hp
```

And again, reprinting the circuit here to cross-reference by the reader:




Fig. 2. Second order BW HP Circuit.

Notice that, instead of scaling the capacitor, per the textbook, we needed to scale the value of the resistors, and that is what we've done.

And we did exactly the same just with the inverting op-amp so that it accepts a scaling coefficient:

```
Listing 3. Inverting Op-Amp netlist.
.SUBCKT gain in out a=1
* K=27.67dB
```

```
Rf      n      out    R=(a*1k)
Ri      in      n      1k
X1      0      n      out    popamp
.ENDS gain
```

and its circuit diagram:




Fig. 3. Inverting Op-amp Circuit.

Notice the reference of a popamp sub-circuit, it is actually a fun way to say "Pseudo-opamp", its netlist came directly from Project 3 and we will also include it below:

```
Listing 4. RC Pseudo-op-amp netlist.
.SUBCKT popamp 4      n      2
Ri      4      n      2MEG
Ei      1      0      4      n      1
R       1      3      1
C       3      0      0.03183
Eo      6      0      3      0      200K
Ro      6      2      75
.ENDS popamp
```

Note that this op-amp accepts three nodes, the first being the non-inverting input, the second the inverting and the last to be the output node. We will not show the pseudo op-amp circuitry in here because it diverts from our purpose of investigating noise reduction system.

The last part of the puzzle is the summing amplifier applied by the band-reject filter of the de-emphasis filter. Here's its netlist:

```
Listing 5. Summing Op-Amp netlist.
.SUBCKT sum in1 in2 out
R1      in1      n      10k
R2      in2      n      10k
Rf      n      out    10k
X1      0      n      out    popamp
.ENDS sum
```

and its circuit diagram:

B. Pre-emphasis

Now we are ready to build our pre-emphasis netlist:




Fig. 4. Summing Op-amp Circuit.

Listing 6. Pre-emphasis netlist.

```
.SUBCKT preemp in out
.PARAM khf='2*530*Pi'
.PARAM klf='w0^2/khf'
Xhp in 1 hp kf=khf c=2n
Xlp 1 2 lp kf=klf r=120k
Xg 2 out gain a='10^(29/20)'
.ENDS preemp
```

its diagram is: Notice how simple this is. We simply called



Fig. 5. Pre-emphasis Op-amp Circuit.

the high pass and low pass filters and scale it to what we want. In here, after numerous attempts to meet specification, we finally settled on scaling the frequency above 500Hz to 530Hz! This is the only way that it met our spec. One may see the entirety of the pre-emphasis filter in appendix A.

If driven by the test current defined TestPreemp.cir a waveform appears:




Fig. 6. Pre-emphasis Op-amp Circuit.

To see the actual data points on the plot, see Appendix C for a larger, and comprehensive view.

C. De-emphasis

Then, we turn our attention to the de-emphasis netlist:

Listing 7. De-emphasis netlist.

```
.SUBCKT deemp in out
.PARAM klf='2*97*Pi'
.PARAM khf='w0^2/klf'
Xlp in 1 lp kf=klf r=120k
Xhp in 2 hp kf=khf c=2n
Xsum 1 2 3 sum
Xg 3 out gain a='10^(3.5/20)'
.ENDS deemp
```

If driven by the test current defined TestDeemp.cir a waveform appears:




Fig. 7. De-emphasis Op-amp Circuit.

We will also present you with a glimpse of its general waveform (again turn to Appendix C for help if you wish to know more.






Fig. 8. Pre-emphasis Op-amp Circuit.

Now we can finally see the result of the NRS system! And we are going to do it by cascading the two emphasis filters and run AC analysis of the entire circuit!

We begin by cascading the output of the pre-emphasis circuit to the input of the de-emphasis circuit. Here's the diagram when we made such connections:



This is the netlist that we used to connect the two filters and at the same time tested the entire system:

Again, see Appendix C for full treatment of data-craving.

Listing 8. NRS netlist.

```
Noise Reduction System – Main
.INCLUDE constants.cir
.INCLUDE popamp.cir
.INCLUDE hp.cir
.INCLUDE lp.cir
.INCLUDE gain.cir
.INCLUDE sum.cir
.INCLUDE preemp.cir
.INCLUDE deemp.cir
Vin      0      1      AC      1
Xpe      1      2      preemp
Xde      2      3      deemp
.AC DEC 40 100 10k
.PRINT VDB(3)
.OPTION list
.OPTION post
.END
```

Notice in this circuit, we are including all of the sub-circuit files that has mutual library dependency for each to function. We have named each file according to their sub-circuit names like the case of writing code in Java. The file constant holds nothing more but these following two constants that is quite crucial to the success of the team. They are:


Listing 9. System constants

```
.PARAM Pi=3.1415926535897932384626433832\\
795028841971693993751058
.PARAM w0='2000*Pi'
```

We have set the value of π and the center frequency at 2000π or 1000Hz. Here's the result of the final wave after passing through our N.R.S.:

APPENDIX A: PRE-EMPHASIS CIRCUIT

Warning: nodes with the same name each has different scope. The repetition of the same name does not mean short. Circuit is made to accomodate each subcircuit that comprised the system.



APPENDIX B: DE-EMPHASIS CIRCUIT

Warning: nodes with the same name each has different scope. The repetition of the same name does not mean short. Circuit is made to accomodate each subcircuit that comprised the system.

